

Stata 12 Tutorial 9

TOPIC: **Estimating and Interpreting Probit Models with Stata: Extensions**

DATA: **mroz.dta** (a *Stata*-format dataset you created in *Stata 12 Tutorial 8*)

TASKS: *Stata 12 Tutorial 9* is an extension of *Stata 12 Tutorial 8*, and therefore deals with the estimation, testing, and interpretation of **probit models** for binary dependent variables. In particular, it illustrates how to use a cross-sectional sample of married women in the United States to investigate whether and how the probability of labour force participation differs between two distinct groups of married women, namely married women who have one or more pre-school aged children and married women who have no pre-school aged children. It demonstrates how *Stata* can be used to conduct an econometric investigation into differences in the conditional probability of labour force participation between these two distinct groups of married women.

- The **Stata commands** that constitute the primary subject of this tutorial are:

probit	Used to compute ML estimates of probit coefficients in probit models of binary dependent variables.
dprobit	Used to compute ML estimates of the marginal probability effects of explanatory variables in probit models.
test	Used after probit estimation to compute Wald tests of linear coefficient equality restrictions on probit coefficients.
lincom	Used after probit estimation to compute and test the marginal effects of individual explanatory variables.
margins	Used after probit estimation to compute estimates of the marginal probability effects of both continuous and categorical (binary) explanatory variables.

- The **Stata statistical functions** used in this tutorial are:

normalden(z)	Computes value of the standard normal <u>density</u> function (p.d.f.) for a given value z of a standard normal random variable.
normal(z)	Computes value of the standard normal <u>distribution</u> function (c.d.f.) for a given value z of a standard normal random variable.
invnormal(p)	Computes the inverse of the standard normal <u>distribution</u> function ; if normal(z) = p , then invnormal(p) = z .

NOTE: *Stata* commands are *case sensitive*. All *Stata* command names must be typed in the Command window in **lower case letters**.

□ **Preparing for Your *Stata* Session**

Before beginning your *Stata* session, use Windows Explorer to copy the *Stata*-format data set **mroz.dta** you created in *Stata 12 Tutorial 8* to the *Stata* **working directory** on the C:-drive or D:-drive of the computer at which you are working.

- **On the computers in Dunning 350**, the default *Stata* working directory is usually **C:\data**.
- **On the computers in MC B111**, the default *Stata* working directory is usually **D:\courses**.

If you did not save the *Stata*-format data set **mroz.dta** you created in *Stata 12 Tutorial 8*, you will have to recreate it from the text-format data file **mroz.raw**, which can be downloaded from the course web site. Consult *Stata 12 Tutorial 8* to refresh your memory on how to do this.

□ **Start Your *Stata* Session**

To start your *Stata* session, double-click on the ***Stata* icon** on the Windows desktop.

After you double-click the ***Stata* icon**, you will see the familiar screen of four *Stata* windows.

□ **Record Your *Stata* Session and *Stata* Commands – log using, cmdlog using**

To record your *Stata* session, including all the *Stata* commands you enter and the results (output) produced by these commands, make a text-format **.log** file named **452tutorial9.log**. To open (begin) the log file **452tutorial9.log**, enter in the Command window:

```
log using 452tutorial9.log
```

This command opens a text-format (ASCII) file called **452tutorial9.log** in the current *Stata* working directory.

Note: It is important to include the **.log** file extension when opening a log file; if you do not, your log file will be in smcl format, a format that only *Stata* can read. Once you have opened the **452tutorial9.log** file, a copy of all the commands you enter during your *Stata* session and of all the results they produce is recorded in that **452tutorial9.log** file.

To record only the *Stata* commands you type during your *Stata* session, use the *Stata* **cmdlog using** command. To start (open) the command log file **452tutorial9.txt**, enter in the Command window:

cmdlog using 452tutorial9

This command opens a plain text-format (ASCII) file called **452tutorial9.txt** in the current *Stata* working directory. All commands you enter during your *Stata* session are recorded in this file.

❑ Loading a *Stata*-Format Dataset into *Stata* – use

In *Stata 12 Tutorial 8*, you created the *Stata*-format dataset **mroz.dta**. If you saved the dataset **mroz.dta** on your own flash memory stick and copied this dataset to the *Stata* working directory before beginning your current *Stata* session, you can simply use the **use** command to read or load **mroz.dta** into memory. If, however, you did not save the *Stata*-format dataset **mroz.dta** you created during *Stata 12 Tutorial 8* and bring it with you on a flash memory stick (or other electronic portable storage device), you will have to repeat most of the section of *Stata 12 Tutorial 8* that explains how to create a *Stata*-format dataset from a text-format data file.

- **To load, or read, into memory the *Stata*-format dataset **mroz.dta****, type in the Command window:

use mroz

This command loads into memory the *Stata*-format dataset **mroz.dta**.

□ Familiarize yourself with the current data set – *describe, summarize*

- **To summarize the contents of the current dataset**, use the **describe** command. Recall from *Stata 12 Tutorial 1* that the **describe** command displays a summary of the contents of the current dataset in memory, which in this case is the *Stata*-format data file **mroz.dta**. Enter the commands:

```
describe, short
describe
```

- **To compute summary statistics for the variables in the current dataset**, use the **summarize** command. Recall from *Stata 12 Tutorial 1* that the **summarize** command computes descriptive summary statistics for all *numeric* variables in the current dataset in memory. Enter the command:

```
summarize
```

- To display summary statistics only for the variables that are used in this tutorial, enter the command:

```
summarize inlf nwifeinc ed exp expsq age dkidslt6
```

Note that the variable **inlf** is a binary variable that takes only the two values 0 and 1. It is the *observed* dependent variable in the probit models estimated in this tutorial.

□ Two Probit Models of Married Women's Participation: Specification

In this section, we consider two different models of married women's labour force participation. Model 2 was introduced in *Stata 12 Tutorial 8*. Model 3 is a generalization of Model 2: it allows all probit coefficients to differ between (1) married women who currently have one or more pre-school aged children and (2) married women who currently have no pre-school aged children.

The ***observed dependent variable*** in both models is the binary variable ***inlf_i*** defined as follows:

$\text{inlf}_i = 1$ if the i -th married woman is in the employed labour force
 $= 0$ if the i -th married woman is not in the employed labour force

The ***explanatory variables*** in Models 2 and 3 are:

- $nwifeinc_i$ = non-wife family income of the i -th woman (in thousands of dollars per year);
 ed_i = years of formal education of the i -th woman (in years);
 exp_i = years of actual work experience of the i -th woman (in years);
 age_i = age of the i -th woman (in years);
 $dkidslt6_i$ = 1 if the i -th woman has one or more children less than 6 years of age, = 0 otherwise.

Four of these explanatory variables – $nwifeinc_i$, ed_i , exp_i , and age_i – are *continuous* variables, whereas the fifth explanatory variable – $dkidslt6_i$ – is a *binary* indicator (dummy) variable.

Note: Refer to *Stata 12 Tutorial 8* to learn how the indicator variable $dkidslt6_i$ is created from the data in the source data file.

Model 2

The **probit index function**, or regression function, **for Model 2** is:

$$x_i^T \beta = \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 dkidslt6_i$$

Remarks: In Model 2, the binary explanatory variable $dkidslt6_i$ enters only additively; only the intercept coefficient in the index function differs between the two groups of married women, those who have pre-school aged children and those who do not.

- ♦ In Model 2, the probit index function for ***married women who have no pre-school aged children***, for whom $dkidslt6_i = 0$, is obtained by setting $dkidslt6_i = 0$ in the index function for Model 2:

$$\begin{aligned}
 (x_i^T \beta | dkidslt6_i = 0) &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 0 \\
 &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i
 \end{aligned}$$

- ♦ In Model 2, the probit index function for ***married women who have one or more pre-school aged children***, for whom $dkidslt6_i = 1$, is obtained by setting $dkidslt6_i = 1$ in the index function for Model 2:

$$\begin{aligned} (x_i^T \beta | dkidslt6_i = 1) &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 1 \\ &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 \end{aligned}$$

- ♦ In Model 2, the **marginal index effect** of the **binary indicator variable $dkidslt6_i$** is simply the difference between (1) the index function for ***married women who currently have one or more pre-school aged children***, $(x_i^T \beta | dkidslt6_i = 1)$ and (2) the index function for ***married women who currently have no pre-school aged children***, $(x_i^T \beta | dkidslt6_i = 0)$:

$$\begin{aligned} &(x_i^T \beta | dkidslt6_i = 1) - (x_i^T \beta | dkidslt6_i = 0) \\ &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 \\ &\quad - (\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i) \\ &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6 \\ &\quad - \beta_0 - \beta_1 nwifeinc_i - \beta_2 ed_i - \beta_3 exp_i - \beta_4 exp_i^2 - \beta_5 age_i \\ &= \beta_6 \end{aligned}$$

- ♦ In Model 2, the **marginal probability effect** of the **binary indicator variable $dkidslt6_i$** is the difference between (1) the conditional probability that $inlf_i = 1$ for ***married women with one or more pre-school aged children*** and (2) the conditional probability that $inlf_i = 1$ for ***married women with no pre-school aged children***:

$$\Pr(inlf_i = 1 | dkidslt6_i = 1) - \Pr(inlf_i = 1 | dkidslt6_i = 0) = \Phi(x_{1i}^T \beta) - \Phi(x_{0i}^T \beta)$$

where $\Phi(*)$ is the cumulative distribution function (c.d.f.) of the standard normal distribution and

$$\mathbf{x}_{li}^T = (1 \text{ nwifeinc}_i \text{ ed}_i \text{ exp}_i \text{ exp}_i^2 \text{ age}_i \text{ 1})$$

$$\mathbf{x}_{oi}^T = (1 \text{ nwifeinc}_i \text{ ed}_i \text{ exp}_i \text{ exp}_i^2 \text{ age}_i \text{ 0})$$

$$\boldsymbol{\beta} = (\beta_0 \ \beta_1 \ \beta_2 \ \beta_3 \ \beta_4 \ \beta_5 \ \beta_6)^T$$

$$\mathbf{x}_{li}^T \boldsymbol{\beta} = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6$$

$$\mathbf{x}_{oi}^T \boldsymbol{\beta} = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i$$

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 1) &= \Phi(\mathbf{x}_{li}^T \boldsymbol{\beta}) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6) \end{aligned}$$

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 0) &= \Phi(\mathbf{x}_{oi}^T \boldsymbol{\beta}) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6 \cdot 0) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i) \end{aligned}$$

Thus, the **marginal probability effect of the indicator variable *dkidslt6_i* in Model 2** is

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 1) - \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 0) &= \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6) \\ &\quad - \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i) \end{aligned}$$

Model 3

The **probit index function**, or regression function, **for Model 3** is:

$$\begin{aligned} \mathbf{x}_i^T \boldsymbol{\beta} &= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ &\quad + \beta_6 \text{dkidslt6}_i + \beta_7 \text{dkidslt6}_i \text{nwifeinc}_i + \beta_8 \text{dkidslt6}_i \text{ed}_i \\ &\quad + \beta_9 \text{dkidslt6}_i \text{exp}_i + \beta_{10} \text{dkidslt6}_i \text{exp}_i^2 + \beta_{11} \text{dkidslt6}_i \text{age}_i \end{aligned}$$

Remarks: Model 3 is the full-interaction generalization of Model 2: it interacts the $dkidslt6_i$ indicator variable with all the other regressors in Model 2, and thereby permits all index function coefficients to differ between the two groups of married women distinguished by $dkidslt6_i$.

- ◆ In Model 3, the probit index function for **married women who currently have no pre-school aged children**, for whom $dkidslt6_i = 0$, is obtained by setting $dkidslt6_i = 0$ in the index function for Model 3:

$$\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid dkidslt6_i = 0 \right) = \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i$$

- ◆ In Model 3, the probit index function for **married women who currently have one or more pre-school aged children**, for whom $dkidslt6_i = 1$, is obtained by setting $dkidslt6_i = 1$ in the index function for Model 3:

$$\begin{aligned} \left(\mathbf{x}_i^T \boldsymbol{\beta} \mid dkidslt6_i = 1 \right) &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i \\ &\quad + \beta_6 1 + \beta_7 1 \cdot nwifeinc_i + \beta_8 1 \cdot ed_i + \beta_9 1 \cdot exp_i + \beta_{10} 1 \cdot exp_i^2 + \beta_{11} 1 \cdot age_i \\ &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i \\ &\quad + \beta_6 + \beta_7 nwifeinc_i + \beta_8 ed_i + \beta_9 exp_i + \beta_{10} exp_i^2 + \beta_{11} age_i \\ &= (\beta_0 + \beta_6) + (\beta_1 + \beta_7) nwifeinc_i + (\beta_2 + \beta_8) ed_i \\ &\quad + (\beta_3 + \beta_9) exp_i + (\beta_4 + \beta_{10}) exp_i^2 + (\beta_5 + \beta_{11}) age_i \end{aligned}$$

- ◆ In Model 3, the **marginal index effect** of the **binary indicator variable** $dkidslt6_i$ is simply the difference between (1) the index function for **married women who currently have one or more pre-school aged children**, $\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid dkidslt6_i = 1 \right)$ and (2) the index function for **married women who currently have no pre-school aged children**, $\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid dkidslt6_i = 0 \right)$:

$$\begin{aligned}
& \left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1 \right) - \left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0 \right) \\
&= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\
&\quad + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \\
&\quad - \left(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \right) \\
&= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\
&\quad + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \\
&\quad - \beta_0 - \beta_1 \text{nwifeinc}_i - \beta_2 \text{ed}_i - \beta_3 \text{exp}_i - \beta_4 \text{exp}_i^2 - \beta_5 \text{age}_i \\
&= \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i
\end{aligned}$$

- ◆ In Model 3, the **marginal probability effect** of the **binary indicator variable** *dkidslt6_i* is the difference between (1) the conditional probability that *inlf_i* = 1 for *married women with one or more pre-school aged children* and (2) the conditional probability that *inlf_i* = 1 for *married women with no pre-school aged children*:

$$\Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1) - \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0) = \Phi(\mathbf{x}_{1i}^T \boldsymbol{\beta}) - \Phi(\mathbf{x}_{0i}^T \boldsymbol{\beta})$$

where $\Phi(*)$ is the cumulative distribution function (c.d.f.) of the standard normal distribution and

$$\mathbf{x}_{1i}^T = (1 \text{ nwifeinc}_i \text{ ed}_i \text{ exp}_i \text{ exp}_i^2 \text{ age}_i \ 1 \text{ nwifeinc}_i \text{ ed}_i \text{ exp}_i \text{ exp}_i^2 \text{ age}_i)$$

$$\mathbf{x}_{0i}^T = (1 \text{ nwifeinc}_i \text{ ed}_i \text{ exp}_i \text{ exp}_i^2 \text{ age}_i \ 0 \ 0 \ 0 \ 0 \ 0 \ 0)$$

$$\boldsymbol{\beta} = (\beta_0 \ \beta_1 \ \beta_2 \ \beta_3 \ \beta_4 \ \beta_5 \ \beta_6 \ \beta_7 \ \beta_8 \ \beta_9 \ \beta_{10} \ \beta_{11})^T$$

$$\begin{aligned}
\mathbf{x}_{1i}^T \boldsymbol{\beta} &= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\
&\quad + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i
\end{aligned}$$

$$\mathbf{x}_{0i}^T \boldsymbol{\beta} = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i$$

$$\begin{aligned}
& \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1) \\
&= \Phi \left(\begin{array}{l} \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \end{array} \right) \\
&= \Phi \left(\begin{array}{l} (\beta_0 + \beta_6) + (\beta_1 + \beta_7) \text{nwifeinc}_i + (\beta_2 + \beta_8) \text{ed}_i \\ + (\beta_3 + \beta_9) \text{exp}_i + (\beta_4 + \beta_{10}) \text{exp}_i^2 + (\beta_5 + \beta_{11}) \text{age}_i \end{array} \right)
\end{aligned}$$

$$\begin{aligned}
& \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0) \\
&= \Phi \left(\begin{array}{l} \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ + \beta_6 0 + \beta_7 0 + \beta_8 0 + \beta_9 0 + \beta_{10} 0 + \beta_{11} 0 \end{array} \right) \\
&= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i)
\end{aligned}$$

Thus, the **marginal probability effect of the indicator variable *dkidslt6_i*** in **Model 3** is

$$\begin{aligned}
& \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1) - \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0) = \\
& \Phi \left(\begin{array}{l} \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \end{array} \right) \\
& - \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i)
\end{aligned}$$

□ Testing the marginal *probability* effect of the binary explanatory variable *dkidslt6_i* -- *test* and *lincom*

Proposition to be Tested

- ♦ Does the conditional probability of labour force participation for married women depend on the presence in the family of one or more dependent children under 6 years of age?
- ♦ Is the probability of labour force participation for married women with given values of nwifeinc_i , ed_i , exp_i , and age_i who currently have one or more pre-school aged children equal to the probability of labour force participation for married women

with the same values of $nwifeinc_i$, ed_i , exp_i , and age_i who currently have no pre-school aged children?

- ♦ Is it true that

$$\begin{aligned} & \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1, \text{nwifeinc}_i, \text{ed}_i, \text{exp}_i, \text{age}_i) \\ &= \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0, \text{nwifeinc}_i, \text{ed}_i, \text{exp}_i, \text{age}_i)? \end{aligned}$$

Null and Alternative Hypotheses: General Formulation

The null hypothesis in general is:

$$H_0: \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1, \dots) = \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0, \dots)$$

The alternative hypothesis in general is:

$$H_1: \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1, \dots) \neq \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0, \dots)$$

Null and Alternative Hypotheses: Model 2

The null hypothesis in general is:

$$H_0: \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1, \dots) = \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0, \dots)$$

For Model 2,

$$\begin{aligned} \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1, \dots) &= \Phi(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6) \\ \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0, \dots) &= \Phi(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i) \end{aligned}$$

These two probabilities are equal if the exclusion restriction $\beta_6 = 0$ is true. In other words, a sufficient condition for these two probabilities to be equal is the exclusion restriction $\beta_6 = 0$.

The **null and alternative hypotheses for Model 2** are therefore:

$$H_0: \beta_6 = 0$$

$$H_1: \beta_6 \neq 0$$

Important Point: A test of the null hypothesis that the **marginal probability effect** of pre-school aged children is zero **is equivalent to** a test of the null hypothesis that the **marginal index effect** of pre-school aged children is zero.

♦ **Marginal probability effect** of pre-school aged children **equals zero** if

$$\Phi(x_i^T \beta | dkidslt6_i = 1) = \Phi(x_i^T \beta | dkidslt6_i = 0).$$

In Model 2,

$$\begin{aligned} \Phi(x_i^T \beta | dkidslt6_i = 1) \\ = \Phi(\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i + \beta_6) \end{aligned}$$

$$\begin{aligned} \Phi(x_i^T \beta | dkidslt6_i = 0) \\ = \Phi(\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i) \end{aligned}$$

Question: What coefficient restriction(s) are sufficient to make these two probabilities equal for any given values of $nwifeinc_i$, ed_i , exp_i , and age_i ?

Answer: By inspection – i.e., by comparing the function $\Phi(x_i^T \beta | dkidslt6_i = 1)$ and the function $\Phi(x_i^T \beta | dkidslt6_i = 0)$ – we can see that a sufficient condition for $\Phi(x_i^T \beta | dkidslt6_i = 1) = \Phi(x_i^T \beta | dkidslt6_i = 0)$ in Model 2 is the single coefficient exclusion restriction $\beta_6 = 0$.

- ♦ **Marginal index effect** of pre-school aged children equals zero if

$$\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1 \right) = \left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0 \right).$$

In Model 2,

$$\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1 \right) = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i + \beta_6$$

$$\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0 \right) = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i$$

Question: What coefficient restriction(s) are sufficient to make these two index functions equal for any given values of nwifeinc_i , ed_i , exp_i , and age_i ?

Answer: By inspection – i.e., by comparing the index function $\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1 \right)$ and the index function $\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0 \right)$ – we can see that a sufficient condition for $\left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 1 \right) = \left(\mathbf{x}_i^T \boldsymbol{\beta} \mid \text{dkidslt6}_i = 0 \right)$ in Model 2 is the single coefficient exclusion restriction $\beta_6 = 0$.

- **Result:** The single coefficient exclusion restriction $\beta_6 = 0$ is sufficient to make the **both the marginal probability effect and the marginal index effect** of pre-school aged children equal to zero in Model 2.

- First, compute ML estimates of probit Model 2 and display the full set of saved results. Enter the following commands:

```
probit inlf nwifeinc ed exp expsq age dkidslt6
ereturn list
```

- To calculate a **Wald test** of H_0 against H_1 and the p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test dkidslt6      or      test dkidslt6 = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[dkidslt6]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent. Note that this **lincom** command merely replicates the test statistic and p-value that are displayed in the output of the **probit** command for the regressor **dkidslt6**.

Null and Alternative Hypotheses: Model 3

The null hypothesis in general is:

$$H_0: \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 1, \dots) = \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 0, \dots)$$

For Model 3,

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 1) &= \Phi(\mathbf{x}_i^T \boldsymbol{\beta} | \text{dkidslt6}_i = 1) \\ &= \Phi \left(\begin{array}{l} \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \end{array} \right) \end{aligned}$$

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 0) &= \Phi(\mathbf{x}_i^T \boldsymbol{\beta} | \text{dkidslt6}_i = 0) \\ &= \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i) \end{aligned}$$

These two probabilities are equal if the six exclusion restrictions $\beta_6 = \beta_7 = \beta_8 = \beta_9 = \beta_{10} = \beta_{11} = 0$ are true. In other words, a sufficient condition for these two probabilities to be equal is the set of six coefficient exclusion restrictions $\beta_j = 0$ for all $j = 6, \dots, 11$.

The **null and alternative hypotheses for Model 3** are therefore:

$$H_0: \beta_j = 0 \quad \forall j = 6, 7, 8, 9, 10, 11$$

$$\Rightarrow \beta_6 = 0 \text{ and } \beta_7 = 0 \text{ and } \beta_8 = 0 \text{ and } \beta_9 = 0 \text{ and } \beta_{10} = 0 \text{ and } \beta_{11} = 0$$

$$H_1: \beta_j \neq 0 \quad j = 6, 7, 8, 9, 10, 11$$

$$\Rightarrow \beta_6 \neq 0 \text{ and/or } \beta_7 \neq 0 \text{ and/or } \beta_8 \neq 0 \text{ and/or } \beta_9 \neq 0 \text{ and/or } \beta_{10} \neq 0 \text{ and/or } \beta_{11} \neq 0$$

Important Point: A test of the null hypothesis that the **marginal probability effect** of pre-school aged children is zero **is equivalent to** a test of the null hypothesis that the **marginal index effect** of pre-school aged children is zero.

- ♦ **Marginal probability effect of pre-school aged children equals zero** if

$$\Phi(x_i^T \beta | dkidslt6_i = 1) = \Phi(x_i^T \beta | dkidslt6_i = 0).$$

In Model 3,

$$\begin{aligned} \Phi(x_i^T \beta | dkidslt6_i = 1) \\ = \Phi \left(\begin{array}{l} \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i \\ + \beta_6 + \beta_7 nwifeinc_i + \beta_8 ed_i + \beta_9 exp_i + \beta_{10} exp_i^2 + \beta_{11} age_i \end{array} \right) \end{aligned}$$

$$\begin{aligned} \Phi(x_i^T \beta | dkidslt6_i = 0) \\ = \Phi(\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i) \end{aligned}$$

Question: What coefficient restriction(s) are sufficient to make these two probabilities equal for any given values of $nwifeinc_i$, ed_i , exp_i , and age_i ?

Answer: By inspection – i.e., by comparing the function $\Phi(x_i^T \beta | dkidslt6_i = 1)$ and the function $\Phi(x_i^T \beta | dkidslt6_i = 0)$ – we can see that a sufficient condition for $\Phi(x_i^T \beta | dkidslt6_i = 1) = \Phi(x_i^T \beta | dkidslt6_i = 0)$ in Model 3 is the set of six coefficient exclusion restrictions $\beta_6 = \beta_7 = \beta_8 = \beta_9 = \beta_{10} = \beta_{11} = 0$.

- ♦ **Marginal index effect of pre-school aged children equals zero** if

$$(x_i^T \beta | dkidslt6_i = 1) = (x_i^T \beta | dkidslt6_i = 0).$$

In Model 3,

$$\begin{aligned} \left(x_i^T \beta \mid dkidslt6_i = 1 \right) &= \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i \\ &\quad + \beta_6 + \beta_7 nwifeinc_i + \beta_8 ed_i + \beta_9 exp_i + \beta_{10} exp_i^2 + \beta_{11} age_i \end{aligned}$$

$$\left(x_i^T \beta \mid dkidslt6_i = 0 \right) = \beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i$$

Question: What coefficient restriction(s) are sufficient to make these two index functions equal for any given values of $nwifeinc_i$, ed_i , exp_i , and age_i ?

Answer: By inspection – i.e., by comparing the index function $\left(x_i^T \beta \mid dkidslt6_i = 1 \right)$ and the index function $\left(x_i^T \beta \mid dkidslt6_i = 0 \right)$ – we can see that a sufficient condition for $\left(x_i^T \beta \mid dkidslt6_i = 1 \right) = \left(x_i^T \beta \mid dkidslt6_i = 0 \right)$ in Model 3 is the set of six coefficient exclusion restrictions $\beta_6 = \beta_7 = \beta_8 = \beta_9 = \beta_{10} = \beta_{11} = 0$.

- **Result:** The six coefficient exclusion restrictions $\beta_6 = \beta_7 = \beta_8 = \beta_9 = \beta_{10} = \beta_{11} = 0$ are sufficient to make the **both the marginal probability effect and the marginal index effect** of pre-school aged children **equal to zero** in Model 3.
- Before estimating Model 3, it is necessary to create the ***dkidslt6_i* interaction variables**. Enter the following **generate** commands:

```
generate d6nwinc = dkidslt6*nwifeinc
generate d6ed = dkidslt6*ed
generate d6exp = dkidslt6*exp
generate d6expsq = dkidslt6*expsq
generate d6age = dkidslt6*age
summarize d6nwinc d6ed d6exp d6expsq d6age
```

- Next, compute ML estimates of probit Model 3 and display the full set of saved results. Enter the following commands:

```
probit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age
ereturn list
```

- To calculate a **Wald test** of H_0 against H_1 and the p-value for the calculated W-statistic, enter the following **test** and **return list** commands:

```
test dkidslt6 d6nwinc d6ed d6exp d6expsq d6age
return list
```

- A second hypothesis test you should perform on Model 3 is a test of the null hypothesis that **all slope coefficient differences** between married women who have one or more pre-school aged children and married women who have no pre-school aged children **equal zero**. The null and alternative hypotheses are:

$$H_0: \beta_j = 0 \quad \forall j = 7, 8, 9, 10, 11$$

$$\Rightarrow \beta_7 = 0 \text{ and } \beta_8 = 0 \text{ and } \beta_9 = 0 \text{ and } \beta_{10} = 0 \text{ and } \beta_{11} = 0$$

$$H_1: \beta_j \neq 0 \quad j = 7, 8, 9, 10, 11$$

$$\Rightarrow \beta_7 \neq 0 \text{ and/or } \beta_8 \neq 0 \text{ and/or } \beta_9 \neq 0 \text{ and/or } \beta_{10} \neq 0 \text{ and/or } \beta_{11} \neq 0$$

Note that the null hypothesis H_0 implies Model 2, whereas the alternative hypothesis H_1 implies Model 3. Enter the **test** command:

```
test d6nwinc d6ed d6exp d6expsq d6age
```

Based on the outcome of this test, would you retain the null hypothesis at the 10 percent significance level? Would you retain the null hypothesis at the 20 percent significance level?

□ Interpreting the coefficient estimates in full-interaction Model 3

Full-interaction Model 3 estimates two distinct sets of probit coefficients: (1) the probit coefficients for married women who have no pre-school aged children (for whom $dkidslt6_i = 0$); and (2) the probit coefficients for married women who have one or more pre-school aged children (for whom $dkidslt6_i = 1$).

- ♦ Recall that the probit index function for Model 3 is:

$$\begin{aligned}
x_i^T \beta &= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\
&\quad + \beta_6 \text{dkidslt6}_i + \beta_7 \text{dkidslt6}_i \text{nwifeinc}_i + \beta_8 \text{dkidslt6}_i \text{ed}_i \\
&\quad + \beta_9 \text{dkidslt6}_i \text{exp}_i + \beta_{10} \text{dkidslt6}_i \text{exp}_i^2 + \beta_{11} \text{dkidslt6}_i \text{age}_i
\end{aligned}$$

- ♦ The **probit index function for married women who have no pre-school aged children** (for whom $\text{dkidslt6}_i = 0$) is obtained by setting the indicator variable $\text{dkidslt6}_i = 0$ in the probit index function for Model 3:

$$(x_i^T \beta | \text{dkidslt6}_i = 0) = \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i$$

Implication: The probit coefficient estimates for married women who have no pre-school aged children (for whom $\text{dkidslt6}_i = 0$) are given directly by the coefficient estimates of the first six terms in the above index function. In particular, **for married women who currently have no pre-school aged children:**

- β_0 = the intercept coefficient for women for whom $\text{dkidslt6}_i = 0$
- β_1 = the slope coefficient of nwifeinc_i for women for whom $\text{dkidslt6}_i = 0$
- β_2 = the slope coefficient of ed_i for women for whom $\text{dkidslt6}_i = 0$
- β_3 = the slope coefficient of exp_i for women for whom $\text{dkidslt6}_i = 0$
- β_4 = the slope coefficient of exp_i^2 for women for whom $\text{dkidslt6}_i = 0$
- β_5 = the slope coefficient of age_i for women for whom $\text{dkidslt6}_i = 0$.

- ♦ The **probit index function for married women who currently have one or more pre-school aged children** (for whom $\text{dkidslt6}_i = 1$) is obtained by setting the indicator variable $\text{dkidslt6}_i = 1$ in the probit index function for Model 3:

$$\begin{aligned}
(x_i^T \beta | \text{dkidslt6}_i = 1) &= \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\
&\quad + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i
\end{aligned}$$

Implication: The probit coefficient estimates for married women who have one or more pre-school aged children (for whom $\text{dkidslt6}_i = 1$) are obtained from Model 3 by summing pairs of coefficient estimates. In particular, **for married women who have one or more pre-school aged children:**

- $\beta_0 + \beta_6$ = the intercept coefficient for women for whom $dkidslt6_i = 1$
- $\beta_1 + \beta_7$ = the slope coefficient of $nwifeinc_i$ for women for whom $dkidslt6_i = 1$
- $\beta_2 + \beta_8$ = the slope coefficient of ed_i for women for whom $dkidslt6_i = 1$
- $\beta_3 + \beta_9$ = the slope coefficient of exp_i for women for whom $dkidslt6_i = 1$
- $\beta_4 + \beta_{10}$ = the slope coefficient of exp_i^2 for women for whom $dkidslt6_i = 1$
- $\beta_5 + \beta_{11}$ = the slope coefficient of age_i for women for whom $dkidslt6_i = 1$.

- To compute from Model 3 the probit coefficient estimates, t-ratios and p-values for those married women who have one or more pre-school aged children (for whom $dkidslt6_i = 1$), enter the following **lincom** commands:

```
lincom _b[_cons] + _b[dkidslt6]
lincom _b[nwifeinc] + _b[d6nwinc]
lincom _b[ed] + _b[d6ed]
lincom _b[exp] + _b[d6exp]
lincom _b[expsq] + _b[d6expsq]
lincom _b[age] + _b[d6age]
```

- Use the estimates of Model 3 to compute a test of the joint significance of all the probit *slope* coefficient estimates for those **married women who have one or more pre-school aged children** (for whom $dkidslt6_i = 1$).

The null and alternative hypotheses are:

$$H_0: \beta_1 + \beta_7 = 0 \text{ and } \beta_2 + \beta_8 = 0 \text{ and } \beta_3 + \beta_9 = 0 \\ \text{and } \beta_4 + \beta_{10} = 0 \text{ and } \beta_5 + \beta_{11} = 0$$

$$H_1: \beta_1 + \beta_7 \neq 0 \text{ and/or } \beta_2 + \beta_8 \neq 0 \text{ and/or } \beta_3 + \beta_9 \neq 0 \\ \text{and/or } \beta_4 + \beta_{10} \neq 0 \text{ and/or } \beta_5 + \beta_{11} \neq 0$$

Enter the following series of linked **test** commands, noting the use of the **notest** and **accumulate** options on the **test** commands:

```
test nwifeinc + d6nwinc = 0, notest
test ed + d6ed = 0, notest accumulate
test exp + d6exp = 0, notest accumulate
test expsq + d6expsq = 0, notest accumulate
test age + d6age = 0, accumulate
```

- Now use the estimates of Model 3 to compute a test of the joint significance of all the probit *slope* coefficient estimates for those **married women who have no pre-school aged children** (for whom $dkidslt6_i = 0$).

The null and alternative hypotheses are:

$$H_0: \beta_1 = 0 \text{ and } \beta_2 = 0 \text{ and } \beta_3 = 0 \text{ and } \beta_4 = 0 \text{ and } \beta_5 = 0$$

$$H_1: \beta_1 \neq 0 \text{ and/or } \beta_2 \neq 0 \text{ and/or } \beta_3 \neq 0 \text{ and/or } \beta_4 \neq 0 \text{ and/or } \beta_5 \neq 0$$

Enter the following **test** command:

```
test nwifeinc ed exp expsq age
```

□ Computing the marginal *probability* effect of the binary explanatory variable $dkidslt6_i$ in Model 3 – *dprobit* with *at(vecname)* option

This section demonstrates how to use the **dprobit** command with the **at(vecname)** option to compute the **marginal *probability* effect of the dummy variable $dkidslt6_i$ in Model 3** for married women who have the **sample *median* values** of the explanatory variables $nwifeinc_i$, ed_i , exp_i , and age_i .

Recall that the **marginal *probability* effect of the dummy variable $dkidslt6_i$ in Model 3** is given by:

$$\begin{aligned} \Pr(\text{inlf}_i = 1 | dkidslt6_i = 1) - \Pr(\text{inlf}_i = 1 | dkidslt6_i = 0) &= \Phi(\mathbf{x}_{1i}^T \boldsymbol{\beta}) - \Phi(\mathbf{x}_{0i}^T \boldsymbol{\beta}) \\ &= \Phi \left(\begin{aligned} &\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i \\ &+ \beta_6 + \beta_7 nwifeinc_i + \beta_8 ed_i + \beta_9 exp_i + \beta_{10} exp_i^2 + \beta_{11} age_i \end{aligned} \right) \\ &\quad - \Phi(\beta_0 + \beta_1 nwifeinc_i + \beta_2 ed_i + \beta_3 exp_i + \beta_4 exp_i^2 + \beta_5 age_i) \end{aligned}$$

The procedure for this computation consists of three steps.

Step 1: Compute an estimate of the probability of labour force participation for married women with the specified characteristics **who currently have one or more dependent children under 6 years of age, for whom $dkidslt6_i = 1$** : i.e., compute an estimate of

$$\Phi(\mathbf{x}_{li}^T \beta) = \Phi \left(\begin{array}{l} \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ + \beta_6 + \beta_7 \text{nwifeinc}_i + \beta_8 \text{ed}_i + \beta_9 \text{exp}_i + \beta_{10} \text{exp}_i^2 + \beta_{11} \text{age}_i \end{array} \right)$$

Step 2: Compute an estimate of the probability of labour force participation for married women with the specified characteristics **who currently have no dependent children under 6 years of age, for whom $\text{dkidslt6}_i = 0$** : i.e., compute an estimate of

$$\Phi(\mathbf{x}_{oi}^T \beta) = \Phi(\beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i)$$

Step 3: Compute an estimate of the difference $\Phi(\mathbf{x}_{li}^T \beta) - \Phi(\mathbf{x}_{oi}^T \beta)$, which is the marginal probability effect of having one or more pre-school aged children for married women who have the specified characteristics.

- Compute (or select) the values of the explanatory variables at which you wish to compute the marginal probability effect of the binary variable dkidslt6_i . For this purpose, we will use the **pooled sample medians** of the explanatory variables nwifeinc_i , ed_i , exp_i , and age_i . Enter the following commands:

```
summarize nwifeinc, detail
return list
scalar nwinc50p = r(p50)
summarize ed, detail
scalar ed50p = r(p50)
summarize exp, detail
scalar exp50p = r(p50)
scalar expsq50p = exp50p^2
summarize age, detail
scalar age50p = r(p50)
scalar list nwinc50p ed50p exp50p expsq50p age50p
```

The sample median values of the explanatory variables computed by these commands are as follows:

```
nwinc50p = 17.700001
ed50p = 12
exp50p = 9
expsq50p = 81
age50p = 43
```

- **Step 1:** Use the **dprobit** command with the **at(vecname)** option to compute the marginal probability effects in Model 3 for median married women whose non-wife family income is \$17,700 per year ($\text{nwifeinc}_i = 17.700$), who have 12 years of formal education ($\text{ed}_i = 12$) and 9 years of actual work experience ($\text{exp}_i = 9$, $\text{expsq}_i = 81$), who are 43 years of age ($\text{age}_i = 43$), and **who have one or more dependent children under 6 years of age ($\text{dkidslt6} = 1$)**. You will first have to create the vector $\mathbf{x}_{ii}^T$ containing the median values of the regressors in Model 3 when $\text{dkidslt6}_i = 1$, since the **dprobit** command does not permit number lists in the **at()** option.

Remember that *Stata* places the equation intercept coefficient β_0 in the *last*, not the first, element of the probit coefficient vector β , so that the coefficient vector β for Model 3 is written in *Stata* format as:

$$\beta = (\beta_1 \ \beta_2 \ \beta_3 \ \beta_4 \ \beta_5 \ \beta_6 \ \beta_7 \ \beta_8 \ \beta_9 \ \beta_{10} \ \beta_{11} \ \beta_0)^T$$

In *Stata* format, the vector $\mathbf{x}_{ii}^T$ for Model 3 thus takes the form:

$$\begin{aligned} \mathbf{x}_{ii}^T &= (\text{nwifeinc}_i \ \text{ed}_i \ \text{exp}_i \ \text{exp}_i^2 \ \text{age}_i \ 1 \ \text{nwifeinc}_i \ \text{ed}_i \ \text{exp}_i \ \text{exp}_i^2 \ \text{age}_i \ 1) \\ &= \begin{pmatrix} \text{nwinc50p} \ \text{ed50p} \ \text{exp50p} \ \text{expsq50p} \ \text{age50p} \ 1 \\ \text{nwinc50p} \ \text{ed50p} \ \text{exp50p} \ \text{expsq50p} \ \text{age50p} \ 1 \end{pmatrix} \end{aligned}$$

Enter the following commands:

```
matrix x1median = (nwinc50p, ed50p, exp50p, expsq50p, age50p, 1,
nwinc50p, ed50p, exp50p, expsq50p, age50p, 1)
matrix list x1median
dprobit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age, at(x1median)
ereturn list
```

Display and save the value of $\Phi(\mathbf{x}_{ii}^T \hat{\beta})$ generated by the above **dprobit** command, where $\Phi(\mathbf{x}_{ii}^T \hat{\beta})$ is an estimate of $\Pr(\text{inlf}_i = 1 | \text{dkidslt6}_i = 1)$. The value of $\Phi(\mathbf{x}_{ii}^T \hat{\beta})$ is temporarily stored as the scalar **e(at)** following execution of the above **dprobit** command. Enter the commands:

```
display e(at)
scalar PHIx1med = e(at)
scalar list PHIx1med
```

These commands save the value of $\Phi(x_{0i}^T \hat{\beta})$ as the scalar **PHIx1med**.

- **Step 2:** Now use the **dprobit** command with the **at(vecname)** option to compute the marginal probability effects in Model 3 for median married women whose non-wife family income is \$17,700 per year ($nwifeinc_i = 17.700$), who have 12 years of formal education ($ed_i = 12$) and 9 years of actual work experience ($exp_i = 9$, $expsq_i = 81$), who are 43 years of age ($age_i = 43$), and **who have no dependent children under 6 years of age ($dkidslt6 = 0$)**. Again, you will first have to create the vector x_{0i}^T containing the median values of the regressors in Model 3 when $dkidslt6_i = 0$.

In *Stata* format, the vector x_{0i}^T for Model 3 takes the form:

$$\begin{aligned} x_{0i}^T &= (nwifeinc_i \quad ed_i \quad exp_i \quad exp_i^2 \quad age_i \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 1) \\ &= (nwinc50p \quad ed50p \quad exp50p \quad expsq50p \quad age50p \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 1) \end{aligned}$$

Enter the following commands:

```
matrix x0median = (nwinc50p, ed50p, exp50p, expsq50p, age50p, 0,
0, 0, 0, 0, 0, 1)
matrix list x0median
dprobit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age, at(x0median)
ereturn list
```

Display and save the value of $\Phi(x_{0i}^T \hat{\beta})$ generated by the above **dprobit** command, where $\Phi(x_{0i}^T \hat{\beta})$ is an estimate of $\Pr(\text{inlf}_i = 1 \mid dkidslt6_i = 0)$. The value of $\Phi(x_{0i}^T \hat{\beta})$ is temporarily stored as the scalar **e(at)** following execution of the above **dprobit** command. Enter the commands:

```
display e(at)
scalar PHIx0med = e(at)
scalar list PHIx0med
```

These commands save the value of $\Phi(x_{0i}^T \hat{\beta})$ as the scalar **PHIx0med**.

- **Step 3:** Finally, compute the estimate of the difference $\Phi(x_{1i}^T \beta) - \Phi(x_{0i}^T \beta)$, which is the marginal probability effect having one or more dependent children under 6 years of age for married women who have the specified characteristics. Enter the commands:

```
scalar diffPHImed = PHIx1med - PHIx0med
scalar list PHIx1med PHIx0med diffPHImed
```

The value of the scalar **diffPHImed** is the estimate for Model 3 of

$$\Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 1) - \Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0) = \Phi(x_{1i}^T \beta) - \Phi(x_{0i}^T \beta)$$

i.e., of the **marginal probability effect of having one or more dependent children under 6 years of age** for married women who have the median characteristics of women in the full sample.

- **Marginal probability effects of continuous explanatory variables in Model 3 -- dprobit**

Background

- ♦ The **marginal probability effects of continuous explanatory variables in probit models** are the partial derivatives of the standard normal c.d.f. $\Phi(x_i^T \beta)$ with respect to the individual explanatory variables:

$$\text{marginal probability effect of } X_j = \frac{\partial \Phi(x_i^T \beta)}{\partial X_{ij}} = \frac{\partial \Phi(x_i^T \beta)}{\partial x_i^T \beta} \frac{\partial x_i^T \beta}{\partial X_{ij}} = \phi(x_i^T \beta) \frac{\partial x_i^T \beta}{\partial X_{ij}}$$

where

$\phi(x_i^T \beta)$ = the value of the standard normal p.d.f. evaluated at $x_i^T \beta$

$\frac{\partial x_i^T \beta}{\partial X_{ij}}$ = the **marginal index effect** of the continuous variable X_j .

- Recall that the **probit index function for Model 3** is:

$$\begin{aligned} \mathbf{x}_i^T \boldsymbol{\beta} = & \beta_0 + \beta_1 \text{nwifeinc}_i + \beta_2 \text{ed}_i + \beta_3 \text{exp}_i + \beta_4 \text{exp}_i^2 + \beta_5 \text{age}_i \\ & + \beta_6 \text{dkidslt6}_i + \beta_7 \text{dkidslt6}_i \text{nwifeinc}_i + \beta_8 \text{dkidslt6}_i \text{ed}_i \\ & + \beta_9 \text{dkidslt6}_i \text{exp}_i + \beta_{10} \text{dkidslt6}_i \text{exp}_i^2 + \beta_{11} \text{dkidslt6}_i \text{age}_i \end{aligned}$$

Marginal Index Effects of Continuous Explanatory Variables – Model 3

- For Model 3, there are **two sets of marginal index effects**, one for women who currently have no pre-school aged children (for whom $\text{dkidslt6}_i = 0$), and the other for women who currently have one or more pre-school aged children (for whom $\text{dkidslt6}_i = 1$).
- The **marginal index effects** of the continuous explanatory variables in Model 3 are obtained by partially differentiating the index function $\mathbf{x}_i^T \boldsymbol{\beta}$ for Model 3 with respect to each of the four continuous explanatory variables nwifeinc_i , ed_i , exp_i , and age_i :

- marginal index effect of $\text{nwifeinc}_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{nwifeinc}_i} = \beta_1 + \beta_7 \text{dkidslt6}_i$
- marginal index effect of $\text{ed}_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{ed}_i} = \beta_2 + \beta_8 \text{dkidslt6}_i$
- marginal index effect of $\text{exp}_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{exp}_i} = \beta_3 + 2\beta_4 \text{exp}_i + (\beta_9 + 2\beta_{10} \text{exp}_i) \text{dkidslt6}_i$
- marginal index effect of $\text{age}_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{age}_i} = \beta_5 + \beta_{11} \text{dkidslt6}_i$

Note: Each of these marginal *index* effects differs depending on whether $\text{dkidslt6}_i = 0$ or $\text{dkidslt6}_i = 1$.

- ♦ The **marginal index effects for married women who currently have *no pre-school aged children*** are obtained by setting the indicator variable $dkidslt6_i = 0$ in expressions 1 to 4 above:

$$5. \text{ marginal index effect of } nwifinc_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial nwifinc_i} = \beta_1$$

$$6. \text{ marginal index effect of } ed_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial ed_i} = \beta_2$$

$$7. \text{ marginal index effect of } exp_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial exp_i} = \beta_3 + 2\beta_4 exp_i$$

$$8. \text{ marginal index effect of } age_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial age_i} = \beta_5$$

- ♦ The **marginal index effects for married women who currently have *one or more pre-school aged children*** are obtained by setting the indicator variable $dkidslt6_i = 1$ in expressions 1 to 4 above:

$$9. \text{ marginal index effect of } nwifinc_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial nwifinc_i} = \beta_1 + \beta_7$$

$$10. \text{ marginal index effect of } ed_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial ed_i} = \beta_2 + \beta_8$$

$$\begin{aligned} 11. \text{ marginal index effect of } exp_i &= \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial exp_i} = \beta_3 + 2\beta_4 exp_i + (\beta_9 + 2\beta_{10} exp_i) \\ &= \beta_3 + \beta_9 + 2(\beta_4 + \beta_{10}) exp_i \end{aligned}$$

$$12. \text{ marginal index effect of } age_i = \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial age_i} = \beta_5 + \beta_{11}$$

Marginal Probability Effects of Continuous Explanatory Variables – Model 3

- ♦ The **marginal probability effects** of the four continuous explanatory variables in Model 3 are:

$$\begin{aligned} 1. \text{ marginal probability effect of } \text{nwifinc}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{nwifinc}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{nwifinc}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta})(\beta_1 + \beta_7 \text{dkidslt6}_i) \end{aligned}$$

$$\begin{aligned} 2. \text{ marginal probability effect of } \text{ed}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{ed}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{ed}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta})(\beta_2 + \beta_8 \text{dkidslt6}_i) \end{aligned}$$

$$\begin{aligned} 3. \text{ marginal probability effect of } \text{exp}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{exp}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{exp}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta})(\beta_3 + 2\beta_4 \text{exp}_i + (\beta_9 + 2\beta_{10} \text{exp}_i) \text{dkidslt6}_i) \end{aligned}$$

$$\begin{aligned} 4. \text{ marginal probability effect of } \text{age}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{age}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{age}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta})(\beta_5 + \beta_{11} \text{dkidslt6}_i) \end{aligned}$$

Notes: There are three features of these marginal probability effects for Model 3 that you should recognize.

- Each of these marginal *probability* effects differs depending on whether $\text{dkidslt6}_i = 0$ or $\text{dkidslt6}_i = 1$.
- The marginal probability effect of a *continuous* explanatory variable X_j is proportional to the marginal index effect of X_j , where the factor of proportionality is the standard normal p.d.f. at $\mathbf{x}_i^T \boldsymbol{\beta}$:

$$\text{marginal probability effect of } X_j = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \times \text{marginal index effect of } X_j$$

3. Estimation of the marginal probability effects of a continuous explanatory variable X_j requires one to choose a specific vector of regressor values $\mathbf{x}_i^T$.

Common choices for $\mathbf{x}_i^T$ are the **sample mean** and **sample median** values of the regressors.

- ♦ The **marginal probability effects for married women who currently have no pre-school aged children** are obtained by setting the indicator variable $\text{dkidslt6}_i = 0$ in expressions 1 to 4 above:

$$\begin{aligned} 5. \text{ marginal probability effect of } \text{nwifcinc}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{nwifcinc}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{nwifcinc}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \beta_1 \end{aligned}$$

$$\begin{aligned} 6. \text{ marginal probability effect of } \text{ed}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{ed}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{ed}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \beta_2 \end{aligned}$$

$$\begin{aligned} 7. \text{ marginal probability effect of } \text{exp}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{exp}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{exp}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta}) (\beta_3 + 2\beta_4 \text{exp}_i) \end{aligned}$$

$$\begin{aligned} 8. \text{ marginal probability effect of } \text{age}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{age}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{age}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \beta_5 \end{aligned}$$

- ♦ The **marginal probability effects for married women who currently have one or more pre-school aged children** are obtained by setting the indicator variable $\text{dkidslt6}_i = 1$ in expressions 1 to 4 above:

$$\begin{aligned} 9. \text{ marginal probability effect of } \text{nwifcinc}_i &= \frac{\partial \Phi(\mathbf{x}_i^T \boldsymbol{\beta})}{\partial \text{nwifcinc}_i} = \phi(\mathbf{x}_i^T \boldsymbol{\beta}) \frac{\partial \mathbf{x}_i^T \boldsymbol{\beta}}{\partial \text{nwifcinc}_i} \\ &= \phi(\mathbf{x}_i^T \boldsymbol{\beta}) (\beta_1 + \beta_7) \end{aligned}$$

$$\begin{aligned}
 10. \text{ marginal probability effect of } ed_i &= \frac{\partial \Phi(x_i^T \beta)}{\partial ed_i} = \phi(x_i^T \beta) \frac{\partial x_i^T \beta}{\partial ed_i} \\
 &= \phi(x_i^T \beta) (\beta_2 + \beta_8)
 \end{aligned}$$

$$\begin{aligned}
 11. \text{ marginal probability effect of } exp_i &= \frac{\partial \Phi(x_i^T \beta)}{\partial exp_i} = \phi(x_i^T \beta) \frac{\partial x_i^T \beta}{\partial exp_i} \\
 &= \phi(x_i^T \beta) (\beta_3 + 2\beta_4 exp_i + \beta_9 + 2\beta_{10} exp_i) \\
 &= \phi(x_i^T \beta) (\beta_3 + \beta_9 + 2(\beta_4 + \beta_{10}) exp_i)
 \end{aligned}$$

$$\begin{aligned}
 12. \text{ marginal probability effect of } age_i &= \frac{\partial \Phi(x_i^T \beta)}{\partial age_i} = \phi(x_i^T \beta) \frac{\partial x_i^T \beta}{\partial age_i} \\
 &= \phi(x_i^T \beta) (\beta_5 + \beta_{11})
 \end{aligned}$$

□ **Testing for zero marginal *probability* effects of *continuous* explanatory variables in Model 3 – *dprobit***

In Model 3, we want to test the proposition that the marginal effect of each continuous explanatory variable on the probability of married women's labour force participation is equal to zero. But since Model 3 incorporates different models of labour force participation for married women who have one or more pre-school aged children and married women who have no pre-school aged children, we will want to test each of these propositions for both groups of married women.

Important Point:

The marginal probability effect of a continuous explanatory variable X_j is proportional to the marginal index effect of X_j , where the factor of proportionality is the standard normal p.d.f. at $x_i^T \beta$:

$$\text{marginal probability effect of } X_j = \phi(x_i^T \beta) \times \text{marginal index effect of } X_j$$

Implication: Any set of coefficient restrictions that is sufficient to make the **marginal index effect** of a continuous explanatory variable **equal to zero** is also sufficient to make the **marginal probability effect** of that continuous explanatory variable **equal to zero**. In other words, testing the null hypothesis that the marginal *index* effect of a continuous explanatory variable equals zero is equivalent to testing the null hypothesis that the marginal *probability* effect of that continuous explanatory variable equals zero.

This section demonstrates how to test for zero marginal probability effects (and zero marginal index effects) of each continuous explanatory variable in Model 3 for each of the two groups of married women: (1) **married women who currently have no pre-school aged children**, for whom the indicator variable $dkidslt6_i = 0$; and (2) **married women who currently have one or more pre-school aged children**, for whom the indicator variable $dkidslt6_i = 1$.

- First, re-estimate probit Model 3. Enter the **probit** command:

```
probit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age
```

- ♦ **Test 1 - Model 3: for married women with no pre-school aged children**
- ♦ **Proposition:** The non-wife income of the family has no effect on the probability of labour force participation for married women who have no pre-school aged children; the marginal probability (and index) effect of $nwifeinc_i$ equals zero for married women for whom $dkidslt6_i = 0$.
- ♦ **For married women for whom $dkidslt6_i = 0$:**

$$\text{marginal probability effect of } nwifeinc_i = \phi(x_i^T \beta) \beta_1$$

A sufficient condition for the marginal probability effect of $nwifeinc_i$ to equal zero for any given values of the regressors x_i^T is $\beta_1 = 0$.

- ♦ **Null and Alternative Hypotheses:**

$$H_0: \beta_1 = 0$$

$$H_1: \beta_1 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test nwifeinc      or      test nwifeinc = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[nwifeinc]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

- ♦ **Test 1 - Model 3: for married women with one or more pre-school aged children**
- ♦ **Proposition:** The non-wife income of the family has no effect on the probability of labour force participation for married women who have one or more pre-school aged children; the marginal probability (and index) effect of $nwifeinc_i$ equals zero for married women for whom $dkidslt6_i = 1$.
- ♦ **For married women for whom $dkidslt6_i = 1$:**

$$\text{marginal probability effect of } nwifeinc_i = \phi(x_i^T \beta)(\beta_1 + \beta_7)$$

A minimally sufficient condition for the marginal probability effect of $nwifeinc_i$ to equal zero for any given values of the regressors x_i^T is $\beta_1 + \beta_7 = 0$.

- ♦ **Null and Alternative Hypotheses:**

$$H_0: \beta_1 + \beta_7 = 0$$

$$H_1: \beta_1 + \beta_7 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test nwifeinc + d6nwinc = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[nwifeinc] + _b[d6nwinc]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

- ♦ ***Test 2 - Model 3: for married women with no pre-school aged children***
- ♦ ***Proposition:*** For married women who have no pre-school aged children, the probability of labour force participation does not depend on their education; the marginal probability (and index) effect of ed_i equals zero for married women for whom $dkidslt6_i = 0$.
- ♦ ***For married women for whom $dkidslt6_i = 0$:***

$$\text{marginal probability effect of } ed_i = \phi(x_i^T \beta) \beta_2$$

A sufficient condition for the marginal probability effect of ed_i to equal zero for any given values of the regressors x_i^T is $\beta_2 = 0$.

- ♦ ***Null and Alternative Hypotheses:***

$$H_0: \beta_2 = 0$$

$$H_1: \beta_2 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test ed      or      test ed = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[ed]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

- ♦ **Test 2 - Model 3: for married women with one or more pre-school aged children**
- ♦ **Proposition:** For married women who have one or more pre-school aged children, the probability of labour force participation does not depend on their education; the marginal probability (and index) effect of ed_i equals zero for married women for whom $dkidslt6_i = 1$.
- ♦ **For married women for whom $dkidslt6_i = 1$:**

$$\text{marginal probability effect of } ed_i = \phi(x_i^T \beta)(\beta_2 + \beta_8)$$

A minimally sufficient condition for the marginal probability effect of ed_i to equal zero for any given values of the regressors x_i^T is $\beta_2 + \beta_8 = 0$.

- ♦ **Null and Alternative Hypotheses:**

$$H_0: \beta_2 + \beta_8 = 0$$

$$H_1: \beta_2 + \beta_8 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test ed + d6ed = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[ed] + _b[d6ed]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

♦ **Test 3 - Model 3: for married women with no pre-school aged children**

- ♦ **Proposition:** Years of actual work experience have no effect on the probability of labour force participation for married women who have no pre-school aged children; the marginal probability (and index) effect of exp_i equals zero for married women for whom $\text{dkidslt6}_i = 0$.

♦ **For married women for whom $\text{dkidslt6}_i = 0$:**

$$\text{marginal probability effect of } \text{exp}_i = \phi(\mathbf{x}_i^T \boldsymbol{\beta})(\beta_3 + 2\beta_4 \text{exp}_i)$$

A sufficient condition for the marginal probability effect of exp_i to equal zero for any given values of the regressors $\mathbf{x}_i^T$ is $\beta_3 = 0$ and $\beta_4 = 0$.

♦ **Null and Alternative Hypotheses:**

$$H_0: \beta_3 = 0 \text{ and } \beta_4 = 0$$

$$H_1: \beta_3 \neq 0 \text{ and/or } \beta_4 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, and **return list** commands:

```
test exp expsq
return list
```

♦ **Test 3 - Model 3: for married women with one or more pre-school aged children**

- ♦ **Proposition:** Years of actual work experience have no effect on the probability of labour force participation for married women who have one or more pre-school aged children; the marginal probability (and index) effect of exp_i equals zero for married women for whom $\text{dkidslt6}_i = 1$.

♦ ***For married women for whom $dkidslt6_i = 1$:***

$$\text{marginal probability effect of } \exp_i = \phi(x_i^T \beta) (\beta_3 + \beta_9 + 2(\beta_4 + \beta_{10}) \exp_i)$$

A minimally sufficient condition for the marginal probability effect of $\exp_i$ to equal zero for any given values of the regressors x_i^T is $\beta_3 + \beta_9 = 0$ and $\beta_4 + \beta_{10} = 0$.

♦ ***Null and Alternative Hypotheses:***

$$H_0: \beta_3 + \beta_9 = 0 \text{ and } \beta_4 + \beta_{10} = 0$$

$$H_1: \beta_3 + \beta_9 \neq 0 \text{ and/or } \beta_4 + \beta_{10} \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test** and **return list** commands:

```
test exp + d6exp = 0, notest
test expsq + d6expsq = 0, accumulate
return list
```

♦ ***Test 4 - Model 3: for married women with no pre-school aged children***

- ♦ ***Proposition:*** For married women who have no pre-school aged children, their age has no effect on their probability of labour force participation; the marginal probability (and index) effect of age_i equals zero for married women for whom $dkidslt6_i = 0$.

♦ ***For married women for whom $dkidslt6_i = 0$:***

$$\text{marginal probability effect of } \text{age}_i = \phi(x_i^T \beta) \beta_5$$

A sufficient condition for the marginal probability effect of age_i to equal zero for any given values of the regressors x_i^T is $\beta_5 = 0$.

♦ ***Null and Alternative Hypotheses:***

$$H_0: \beta_5 = 0$$

$$H_1: \beta_5 \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test age          or      test age = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[age]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

- ♦ ***Test 4 - Model 3: for married women with one or more pre-school aged children***
- ♦ ***Proposition:*** For married women who have one or more pre-school aged children, their age has no effect on their probability of labour force participation; the marginal probability (and index) effect of age_i equals zero for married women for whom $dkidslt6_i = 1$.
- ♦ ***For married women for whom $dkidslt6_i = 1$:***

$$\text{marginal probability effect of } age_i = \phi(x_i^T \beta)(\beta_5 + \beta_{11})$$

A minimally sufficient condition for the marginal probability effect of age_i to equal zero for any given values of the regressors x_i^T is $\beta_5 + \beta_{11} = 0$.

- ♦ ***Null and Alternative Hypotheses:***

$$H_0: \beta_5 + \beta_{11} = 0$$

$$H_1: \beta_5 + \beta_{11} \neq 0$$

- To calculate a **Wald test** of this hypothesis and the corresponding p-value for the calculated W-statistic, enter the following **test**, **return list** and **display** commands:

```
test age + d6age = 0
return list
display sqrt(r(chi2))
```

- To calculate a **two-tail asymptotic t-test** of H_0 against H_1 , enter the following **lincom**, **return list** and **display** commands:

```
lincom _b[age] + _b[d6age]
return list
display r(estimate)/r(se)
```

Compare the results of this two-tail t-test with those of the previous Wald test. You should be able to explain why the two tests are equivalent.

☐ **Computing estimates of the marginal *probability* effects of *continuous* explanatory variables in Model 3 -- *dprobit***

Introduction

For any explanatory variable, there are **two distinct *empirical* questions** that an econometric investigation of married women's labour force participation (or any other binary outcome) should address.

- ♦ The first question concerns the ***existence*** of a relationship: is a particular explanatory variable related to the probability of married women's labour force participation, conditional on other explanatory variables included in the model? In other words, is the marginal probability effect of a particular explanatory variable on the probability of married women's labour force participation non-zero?
- ♦ The second question concerns the ***magnitude*** of the relationship: how large a change in the conditional probability of married women's labour force participation is associated with a one-unit increase in the value of a particular continuous explanatory variable, holding constant the values of all other explanatory variables included in the model?

The previous section addressed the first question for each of the four continuous explanatory variables in Model 3. This section demonstrates how to address the second question for each of the continuous explanatory variables $nwifeinc_i$, ed_i , exp_i , and age_i .

Procedure

Recall that the **marginal probability effect of a continuous explanatory variable X_j** is proportional to the marginal index effect of X_j , where the factor of proportionality is the standard normal p.d.f. evaluated at $x_i^T \beta$:

$$\text{marginal probability effect of } X_j = \phi(x_i^T \beta) \times \text{marginal index effect of } X_j$$

This expression implies that to compute estimates of the marginal *probability* effect of each *continuous* explanatory variable, we must first do two things. First, we must compute an estimate $x_i^T \hat{\beta}$ of $x_i^T \beta$. Second, we must compute the value of $\phi(x_i^T \hat{\beta})$, i.e., the value of the standard normal density function evaluated at $x_i^T \hat{\beta}$.

Marginal probability effects for married women for whom $dkidslt6_i = 0$

In this section, we compute the marginal probability effects of the four continuous explanatory variables in Model 3 for married women who have the sample median values of $nwifeinc_i$, ed_i , exp_i , and age_i , and **no pre-schooled aged children** (for whom $dkidslt6_i = 0$).

- To compute marginal probability effects for the median married woman who has no pre-school aged children, we first re-estimate Model 3 using the **dprobit** command with the **at(vecname)** option. The vector to use in the **at(vecname)** option is the vector x_{0i}^T containing the median values of the regressors in Model 3 when $dkidslt6_i = 0$:

$$\begin{aligned} x_{0i}^T &= (nwifeinc_i \quad ed_i \quad exp_i \quad exp_i^2 \quad age_i \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 1) \\ &= (nwinc50p \quad ed50p \quad exp50p \quad expsq50p \quad age50p \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 0 \quad 1) \end{aligned}$$

You previously created the vector x_{0i}^T and named it **x0median**. So simply enter the commands:

```
dprobit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age, at(x0median)
ereturn list
display e(at)
```

Recall that the scalar **e(at)** contains the value of $\Phi(x_{0i}^T \hat{\beta})$ generated by the previous **dprobit** command, where $\Phi(x_{0i}^T \hat{\beta})$ is an estimate of $\Pr(\text{inlf}_i = 1 \mid \text{dkidslt6}_i = 0)$.

- Second, use the *Stata* statistical function **invnormal()** to save the value of $x_{0i}^T \hat{\beta}$. Enter the commands:

```
scalar x0medbhat = invnormal(e(at))
scalar list x0medbhat
```

- Third, use the *Stata* statistical function **normalden()** to save as a scalar the value of $\phi(x_{0i}^T \hat{\beta})$, which is the standard normal density function (or p.d.f.) evaluated at $x_{0i}^T \hat{\beta}$. Enter the commands:

```
scalar phix0med = normalden(x0medbhat)
scalar list phix0med
```

These commands save the value of $\phi(x_{0i}^T \hat{\beta})$ as the scalar **phix0med**.

- Compute the **estimated marginal probability effect of explanatory variable *nwifeinc_i*** for the **median married woman who has no pre-school aged children**, which when **dkidslt6_i = 0** is given by the function:

$$\text{estimated marginal probability effect of } nwifeinc_i = \phi(x_{0i}^T \hat{\beta}) \hat{\beta}_1$$

Enter the **lincom** command:

```
lincom phix0med*_b[nwifeinc]
```

- Compute the **estimated marginal probability effect of explanatory variable ed_i** for the **median married woman who has no pre-school aged children**, which when **$dkidslt6_i = 0$** is given by the function:

$$\text{estimated marginal probability effect of } ed_i = \phi(x_{0i}^T \hat{\beta}) \hat{\beta}_2$$

Enter the **lincom** command:

```
lincom phix0med*_b[ed]
```

- Compute the **estimated marginal probability effect of explanatory variable exp_i** for the **median married woman who has no pre-school aged children**, which when **$dkidslt6_i = 0$** is given by the function:

$$\text{estimated marginal probability effect of } exp_i = \phi(x_{0i}^T \hat{\beta}) (\hat{\beta}_3 + 2 \hat{\beta}_4 \exp50p)$$

Enter the **lincom** command:

```
lincom phix0med*(_b[exp] + 2*_b[expsq]*exp50p)
```

- Compute the **estimated marginal probability effect of explanatory variable age_i** for the **median married woman who has no pre-school aged children**, which when **$dkidslt6_i = 0$** is given by the function:

$$\text{estimated marginal probability effect of } age_i = \phi(x_{0i}^T \hat{\beta}) \hat{\beta}_5$$

Enter the **lincom** command:

```
lincom phix0med*_b[age]
```

Marginal probability effects for married women for whom $dkidslt6_i = 1$

In this section, we compute the marginal probability effects of the four continuous explanatory variables in Model 3 for married women who have the sample median values of $nwifeinc_i$, ed_i , exp_i , and age_i , and **one or more pre-schooled aged children** (for whom $dkidslt6_i = 1$).

- To compute marginal probability effects for the median married woman who has at least one pre-school aged child, we first re-estimate Model 3 using the **dprobit** command with the **at(vecname)** option. The vector to use in the **at(vecname)** option is the vector x_{ii}^T containing the median values of the regressors in Model 3 when $dkidslt6_i = 1$:

$$\begin{aligned} x_{ii}^T &= (nwifeinc_i \quad ed_i \quad exp_i \quad exp_i^2 \quad age_i \quad 1 \quad nwifeinc_i \quad ed_i \quad exp_i \quad exp_i^2 \quad age_i \quad 1) \\ &= \begin{pmatrix} nwinc50p & ed50p & exp50p & expsq50p & age50p & 1 \\ nwinc50p & ed50p & exp50p & expsq50p & age50p & 1 \end{pmatrix} \end{aligned}$$

You previously created the vector x_{ii}^T and named it **x1median**. So simply enter the commands:

```
dprobit inlf nwifeinc ed exp expsq age dkidslt6 d6nwinc d6ed
d6exp d6expsq d6age, at(x1median)
ereturn list
display e(at)
```

Recall that the scalar **e(at)** contains the value of $\Phi(x_{ii}^T \hat{\beta})$ generated by the previous **dprobit** command, where $\Phi(x_{ii}^T \hat{\beta})$ is an estimate of $\Pr(\text{inlf}_i = 1 \mid dkidslt6_i = 1)$.

- Second, use the *Stata* statistical function **invnormal()** to save the value of $x_{ii}^T \hat{\beta}$. Enter the commands:

```
scalar x1medbhat = invnormal(e(at))
scalar list x1medbhat
```

- Third, use the *Stata* statistical function **normalden()** to save as a scalar the value of $\phi(\mathbf{x}_{li}^T \hat{\beta})$, which is the standard normal density function (or p.d.f.) evaluated at $\mathbf{x}_{li}^T \hat{\beta}$. Enter the commands:

```
scalar phix1med = normalden(x1medbhat)
scalar list phix1med
```

These commands save the value of $\phi(\mathbf{x}_{li}^T \hat{\beta})$ as the scalar **phix1med**.

- Compute the **estimated marginal probability effect of explanatory variable *nwifeinc_i*** for the **median married woman who has one or more pre-school aged children**, which when **dkidslt6_i = 1** is given by the function:

$$\text{estimated marginal probability effect of } nwifeinc_i = \phi(\mathbf{x}_{li}^T \hat{\beta})(\hat{\beta}_1 + \hat{\beta}_7)$$

Enter the **lincom** command:

```
lincom phix1med*(_b[nwifeinc] + _b[d6nwinc])
```

- Compute the **estimated marginal probability effect of explanatory variable *ed_i*** for the **median married woman who has one or more pre-school aged children**, which when **dkidslt6_i = 1** is given by the function:

$$\text{estimated marginal probability effect of } ed_i = \phi(\mathbf{x}_{li}^T \hat{\beta})(\hat{\beta}_2 + \hat{\beta}_8)$$

Enter the **lincom** command:

```
lincom phix1med*(_b[ed] + _b[d6ed])
```

- Compute the **estimated marginal probability effect of explanatory variable *exp_i*** for the **median married woman who has one or more pre-school aged children**, which when **dkidslt6_i = 1** is given by the function:

$$\text{estimated marginal probability effect of } exp_i = \phi(\mathbf{x}_{li}^T \hat{\beta})(\hat{\beta}_3 + \hat{\beta}_9 + 2(\hat{\beta}_4 + \hat{\beta}_{10}) \exp 50p)$$

Enter *on one line* the **lincom** command:

```
lincom phix1med*(_b[exp] + _b[d6exp] + 2*(_b[expsq] +
_b[d6expsq])*exp50p)
```

- Compute the **estimated marginal probability effect of explanatory variable age_i** for the **median married woman who has one or more pre-school aged children**, which when **dkidslt6_i = 1** is given by the function:

$$\text{estimated marginal probability effect of } age_i = \phi(x_{11}^T \beta)(\beta_5 + \beta_{11})$$

Enter the **lincom** command:

```
lincom phix1med*(_b[age] + _b[d6age])
```

□ Computing the marginal *probability* effect of the binary explanatory variable **dkidslt6_i** in Model 3 – *probit* command followed by *margins* command

You have previously computed an estimate of the marginal probability effect of the **binary explanatory variable $dkidslt6_i$** in Model 3 at the sample median values of the continuous explanatory variables; however, that procedure, while completely correct, was somewhat laborious. This section demonstrates a much shorter and easier procedure that uses the **margins** command after Maximum Likelihood estimation of Model 3 with a **probit** command for computing the **marginal probability effect of the dummy variable $dkidslt6_i$ in Model 3** for married women who have the sample median values of the explanatory variables $nwifeinc_i$, ed_i , exp_i , and age_i .

- First, use the **probit** command to re-estimate Model 3, with all regressors entered in factor-variable notation to distinguish between **continuous and categorical explanatory variables**. Model 3 contains four *continuous* explanatory variables, specifically $nwifeinc_i$, ed_i , exp_i , and age_i , and one *binary categorical* explanatory variable, **dkidslt6_i**. Enter *on one line* the following command:

```
probit inlf c.nwifeinc c.ed c.exp c.exp#c.exp c.age i.dkidslt6
i.dkidslt6#(c.nwifeinc c.ed c.exp c.exp#c.exp c.age)
```

- Second, use a **margins** command with the **at()** option to compute estimates of the **conditional probability of labour force participation** for (1) **married women with no pre-school aged children**, for whom **dkidslt6_i = 0**, and (2) **married women with one or more pre-school aged children**, for whom **dkidslt6_i = 1**. Note that the **at()** option is used tell *Stata* that these conditional probabilities of labour force participation are to be computed at the **sample median values** of the four continuous explanatory variables **nwifeinc_i**, **ed_i**, **exp_i**, and **age_i**. Enter the following **margins** command:

```
margins i.dkidslt6, at((median) nwifeinc ed exp age)
```

- Third, use a second **margins** command with the **at()** option to compute an estimate of the **marginal probability effect of dkidslt6_i**, which by definition is the difference in the conditional probability of labour force participation between married women with pre-school aged children (for whom **dkidslt6_i = 1**) and married women with no pre-school aged children (for whom **dkidslt6_i = 0**). Enter the following two **margins** commands:

```
margins r.dkidslt6, at((median) nwifeinc ed exp age)
margins r.dkidslt6, at((median) nwifeinc ed exp age)
contrast(nowald effects)
```

Note that the first of the above **margins** commands reports a **Wald test** of the null hypothesis that the **marginal probability effect of dkidslt6_i** at sample median values of **nwifeinc_i**, **ed_i**, **exp_i**, and **age_i** is equal to zero, whereas the second **margins** command reports an equivalent **large-sample t-test** of the same null hypothesis. Otherwise, the results produced by these two **margins** commands are identical.

□ Computing the marginal probability effect of the continuous explanatory variables in Model 3 – *probit* command followed by *margins* command

The **margins** command can easily be used to compute for each of the four **continuous explanatory variables** in Model 3 estimates of the **marginal probability effect** of that continuous variable for both **married women without pre-school aged children** and **married women with one or more pre-school aged children**. The **margins** command can also be used to compute the difference between these two marginal effects for each continuous explanatory variable, and to perform a two-tail test of the null hypothesis that this difference is equal to zero. This section shows you how to use the **margins**

command to perform these calculations for each of the four continuous explanatory variables **nwifeinc_i**, **ed_i**, **exp_i**, and **age_i** in Model 3.

1. The continuous explanatory variable **nwifeinc_i**

- First, use a **margins** command to compute estimates of the **marginal probability effect of non-wife family income **nwifeinc_i**** for (1) the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0** and (2) the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1**. Enter *on one line* the **margins** command:

```
margins i.dkidslt6, dydx(c.nwifeinc) at((median) nwifeinc ed exp age)
```

- Second, use a second **margins** command to compute an estimate of the **difference** between (1) the **estimated marginal probability effect of **nwifeinc_i**** for the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1** and (2) the **estimated marginal probability effect of **nwifeinc_i**** for the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0**. Enter *on one line* each of the following two **margins** commands:

```
margins r.dkidslt6, dydx(c.nwifeinc) at((median) nwifeinc ed exp age)
```

```
margins r.dkidslt6, dydx(c.nwifeinc) at((median) nwifeinc ed exp age) contrast(nowald effects)
```

2. The continuous explanatory variable **ed_i**

- First, use a **margins** command to compute estimates of the **marginal probability effect of years of formal education **ed_i**** for (1) the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0** and (2) the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1**. Enter *on one line* the **margins** command:

```
margins i.dkidslt6, dydx(c.ed) at((median) nwifeinc ed exp age)
```

- Second, use a second **margins** command to compute an estimate of the *difference* between (1) the **estimated marginal probability effect of ed_i** for the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1** and (2) the **estimated marginal probability effect of ed_i** for the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0**. Enter *on one line* each of the following two **margins** commands:

```
margins r.dkidslt6, dydx(c.ed) at((median) nwifeinc ed exp age)
margins r.dkidslt6, dydx(c.ed) at((median) nwifeinc ed exp age)
contrast(nowald effects)
```

3. The continuous explanatory variable exp_i

- First, use a **margins** command to compute estimates of the *marginal probability effect of years of work experience exp_i* for (1) the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0** and (2) the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1**. Enter *on one line* the **margins** command:

```
margins i.dkidslt6, dydx(c.exp) at((median) nwifeinc ed exp age)
```

- Second, use a second **margins** command to compute an estimate of the *difference* between (1) the **estimated marginal probability effect of exp_i** for the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1** and (2) the **estimated marginal probability effect of exp_i** for the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0**. Enter *on one line* each of the following two **margins** commands:

```
margins r.dkidslt6, dydx(c.exp) at((median) nwifeinc ed exp age)
margins r.dkidslt6, dydx(c.exp) at((median) nwifeinc ed exp age)
contrast(nowald effects)
```

4. The continuous explanatory variable age_i

- First, use a **margins** command to compute estimates of the *marginal probability effect of age age_i* for (1) the **median married woman who has no pre-school aged children**, for whom **dkidslt6_i = 0** and (2) the **median married woman who has one or more pre-school aged children**, for whom **dkidslt6_i = 1**. Enter *on one line* the **margins** command:

```
margins i.dkidslt6, dydx(c.age) at((median) nwifeinc ed exp age)
```

- Second, use a second **margins** command to compute an estimate of the *difference* between (1) the **estimated marginal probability effect of age_i** for the *median married woman who has one or more pre-school aged children*, for whom **dkidslt6_i = 1** and (2) the **estimated marginal probability effect of age_i** for the *median married woman who has no pre-school aged children*, for whom **dkidslt6_i = 0**. Enter *on one line* each of the following two **margins** commands:

```
margins r.dkidslt6, dydx(c.age) at((median) nwifeinc ed exp age)
margins r.dkidslt6, dydx(c.age) at((median) nwifeinc ed exp age)
contrast(nowald effects)
```

□ Preparing to End Your *Stata* Session

Before you end your *Stata* session, you should do two things.

- First, you will want to **save the current data set**. Enter the following **save** command with the **replace** option to save the current data set as *Stata*-format data set **mroz.dta**:

```
save mroz, replace
```

- Second, **close the command log file** you have been recording. Enter the command:

```
cmdlog close
```

- Third, **close the log file** you have been recording. Enter the command:

```
log close
```

□ End Your *Stata* Session -- exit

- **To end your *Stata* session**, use the **exit** command. Enter the command:

```
exit      or      exit, clear
```

□ Cleaning Up and Clearing Out

After returning to Windows, you should copy all the files you have used and created during your *Stata* session to your own portable electronic storage device, such as a flash memory stick. These files will be found in the ***Stata* working directory**, which is usually **C:\data** on the computers in Dunning 350, and **D:\courses** on the computers in MC B111. There are three files you will want to be sure you have: the *Stata* log file **452tutorial9.log**; the *Stata* command log file **452tutorial9.txt**; and the *Stata*-format data set **mroz.dta**. Use the Windows **copy** command to copy any files you want to keep to your own personal portable storage device (e.g., a flash memory stick).

Finally, **as a courtesy to other users** of the computing classroom, please delete all the files you have used or created from the *Stata* working directory.